Assignment 05

(1) Find the limit of $\frac{\sin x^2}{x}$ as x-> 0

```
Limit[Sin[x^2] / x, x → 0]
```

```
0
```

(2) Find the derivative of

(i) $y(x) = \sin(e^{x^2})$

(ii) $y(x) = \frac{x^{3/4}\sqrt{x^2+1}}{(3x+2)^5}$ [Use Simplify[...] to get the answer in a nice shape]

(iii) $y(x) = x \operatorname{arcsinh}(x/3) - \sqrt{9+x^2}$

(iv) The second derivative of $y(x)=e^x \sin x$

(v) The partial derivative with respect to x of $f = \frac{\sin(xy)}{\cos(x+y)}$

(vi) Find $\frac{\partial^3 u}{\partial y^2 \partial x}$ given $u=\ln(x^2+y)$

```
Print["(i)"];
Clear[y, x];
y = Sin[Exp[x^2]];
D[y, x]

Print[];
Print["(ii)"];
Clear[y, x];
y = x^(3/4) Sqrt[x^2 + 1] / (3 x + 2 )^5;
Simplify[D[y, x]]

Print[];
Print["(iii)"];
Clear[y, x];
y = x ArcSinh[x / 3] - Sqrt[9 + x^2];
Simplify[D[y, x]]

Print[];
Print["(iv)"];
Clear[y, x];
y = Exp[x] Sin[x];
Simplify[D[y, {x, 2}]]

Print[];
Print["(v)"];
Clear[f, y, x];
f =  Sin[x y] / Cos[x + y];
Simplify[D[f, x]]

Print[];
Print["(vi)"];
Clear[u, y, x];
u = Log[x^2 + y];
Simplify[D[u, x,  {y, 2}]]
```

(i)

$$2 e^{x^2} x \operatorname{Cos}\left[e^{x^2}\right]$$

(ii)

$$\frac{6 - 51 x + 14 x^2 - 39 x^3}{4 x^{1/4} (2 + 3 x)^6 \sqrt{1 + x^2}}$$

(iii)

$\text{ArcSinh}\left[\frac{x}{3}\right]$

(iv)

$2\, e^x\, \text{Cos}[x]$

(v)

$\text{Sec}[x+y]\ (y\, \text{Cos}[x\, y] + \text{Sin}[x\, y]\, \text{Tan}[x+y])$

(vi)

$$\frac{4\,x}{\left(x^2+y\right)^3}$$

(3) Maxima and minima

The function f(x)=x+sin(5x) has three relative maxima and two relative minima in the interval [0,Pi].

(i) Plot the function to get an idea of their locations.

(ii) Find them.

```
Clear[f]
f[x_] := x + Sin[5 x];
Print["(i) Plotting the function"];
Plot[f[x], {x, 0, Pi}]
```



```
Print["(iii)"];
Print["Minima"];
FindMinimum[f[x], {x, 1}]
FindMinimum[f[x], {x, 2}]
Print["Maxima"];
FindMaximum[f[x], {x, 0.4}]
FindMaximum[f[x], {x, 1.5}]
FindMaximum[f[x], {x, 3.}]
```

```
(iii)

Minima

{-0.0775897, {x → 0.902206}}

{1.17905, {x → 2.15884}}

Maxima

{1.33423, {x → 0.354431}}

{2.59086, {x → 1.61107}}

{3.8475, {x → 2.8677}}
```

(4) Suppose f(x) = -x^3 + x + 1.

(i) Find f'(x) using a single command from *Mathematica.*

(ii) Find f'(x) using the command "Limit" from *Mathematica.*

(iii) Find f'(1) using item (i) and compare it with results from the definition of derivative obtained with a do-loop. In this case, use increments h to x, which vary as 0.1, 0.01, ...0.000001

(iv) Determine the equation of the line tangent to the graph of f(x) at the point (1,f(1)).

[Hint: the equation of the tangent line to a curve at xo is
f(xo) + f'(xo) (x-xo)]

(v) Plot both f(x) and the tangent line in the interval [-1.5,2].

```
Clear[f, df, slope, ftan];
f[x_] := -x^3 + x + 1
Print["(i) The derivative is"];
df = D[f[x], x]

Print[]
Print["(ii) It can also be found with the Limit command"];
Limit[(f[x + h] - f[x]) / h, h → 0]

Print[]
Print["(iii)"];
Print["f'(1) is the slope of the function at x=1"];
slope = df /. x → 1
Print["This value can also be approached by using a small increment to x:"];
Clear[incr, der];
Do[
  incr = 1. × 10^(-k);
  der = (f[1 + incr] - f[1]) / incr;
  Print["increment ", incr, " gives ", der];
  , {k, 1, 6}];

Print[]
Print["(iv) Tangent to f(x) at point (1,f(1))"];
ftan = f[1] + slope (x - 1)

Print[]
Print["(v) f(x) and its tangent at point (1,f(1))"];
Plot[{f[x], ftan}, {x, -1.5, 2}, PlotStyle → {{Blue, Thick}, {Red, Thick}},
 LabelStyle → Directive[Black, Bold, Medium]]
```

(i) The derivative is

$1 - 3x^2$

(ii) It can also be found with the Limit command

$1 - 3x^2$

(iii)

f'(1) is the slope of the function at x=1

-2

This value can also be approached by using a small increment to x:

increment 0.1 gives -2.31

increment 0.01 gives -2.0301

increment 0.001 gives -2.003

increment 0.0001 gives -2.0003

increment 0.00001 gives -2.00003

increment $1. \times 10^{-6}$ gives -2.

(iv) Tangent to f(x) at point (1,f(1))

$1 - 2 \times (-1 + x)$

(v) f(x) and its tangent at point (1,f(1))



(5) Repeat the last item above, but embed it all within “Manipulate” so that the point of tangency between the line and the cubic function goes continuosuly from x=-1.5 to x=2.0

```
Clear[f, df];
f[x_] := -x^3 + x + 1;
df = D[f[x], x];
Manipulate[
 slope = df /. x → a;
 Plot[{f[x], f[a] + slope (x - a)}, {x, -1.5, 2} ,
  PlotStyle → {{Blue, Thick}, {Red, Thick}}, PlotRange → {-4, 4},
  LabelStyle → Directive[Black, Bold, Medium]], {a, -1.5, 2}]
```

(6) Repeat the exercise above, but now with Animate and this time for the function: f(x)=x*sin(x^2) +1

[This is the same function used in wikipedia to illustrate the notion of derivate: http://en.wikipedia.org/wiki/Derivative].
The plot range should be {-2,4} for x and also for y, whereas the function and the tangent should appear just for x from -1 to 3.

```
Clear[f, df];
f[x_] := x Sin[x^2] + 1;
df = D[f[x], x];
Animate[
 slope = df /. x → a;
 Plot[{f[x], f[a] + slope (x - a)}, {x, -1., 3} ,
  PlotStyle → {{Blue, Thick}, {Red, Thick}}, PlotRange → {{-2, 4}, {-2, 4}},
  LabelStyle → Directive[Black, Bold, Medium]], {a, -1., 3}]
```

(7)
(i) Find the Taylor expansion of $e^{\sin(u)}$ about u=0 to order u^7
(ii) Convert the power series above to an ordinary expression (that is use Normal[...])
(iii) Find the Taylor expansion of $\left(1+x^4\right)^{1/3}$ about x=0 up to order x^4,x^8, and x^{24} and integrate each result for {x, 0, 1}
(iv) Find the integral
$\int_0^1 \left(1+x^4\right)^{1/3} dx$
and compare with the results from item (iii)

```
Print["(i)"];
Clear[u];
Series[Exp[Sin[u]], {u, 0, 7}]

Print[];
Print["(ii)"];
Clear[u];
Normal[Series[Exp[Sin[u]], {u, 0, 7}]]

Print[];
Print["(iii)"];
Clear[t4, t8, t24, x];
Print["4th order"];
t4 = Normal[Series[(1 + x^4)^(1/3), {x, 0, 4}]]
Integrate[t4, {x, 0, 1.}]
Print["8th order"];
t8 = Normal[Series[(1 + x^4)^(1/3), {x, 0, 8}]]
Integrate[t8, {x, 0, 1.}]
Print["24th order"];
t24 = Normal[Series[(1 + x^4)^(1/3), {x, 0, 24}]]
Integrate[t24, {x, 0, 1.}]


Print[];
Print["(iv)"];
NIntegrate[(1 + x^4)^(1/3), {x, 0, 1}]
```

(i)

$1 + u + \frac{u^2}{2} - \frac{u^4}{8} - \frac{u^5}{15} - \frac{u^6}{240} + \frac{u^7}{90} + O[u]^8$

(ii)

$1 + u + \frac{u^2}{2} - \frac{u^4}{8} - \frac{u^5}{15} - \frac{u^6}{240} + \frac{u^7}{90}$

(iii)

4th order

$1 + \frac{x^4}{3}$

1.06667

8th order

$1 + \frac{x^4}{3} - \frac{x^8}{9}$

1.05432

24th order

$1 + \frac{x^4}{3} - \frac{x^8}{9} + \frac{5 x^{12}}{81} - \frac{10 x^{16}}{243} + \frac{22 x^{20}}{729} - \frac{154 x^{24}}{6561}$

1.05715

(iv)

1.05753

I needed to go to a very high order of the Taylor expansion to get a result close to the integration of the actual funtion

(8)

(i) Evaluate $\int_0^1 \frac{4}{1+x^2} \, dx$

(ii) Evaluate the integral and then differentiate the result to recover the integrand [you may need to use Simplify[...]]

$\int \frac{x}{a^3+x^3} \, dx$

(iii) $\int \frac{2x^2-x+4}{x^3+4x} \, dx$

(iv) Assuming that the constant a is both real and positive, evaluate the integral $\int_{-\infty}^{\infty} x^2 e^{-2 a x^2} \, dx$

(v) Assuming that a>0 and m≥0, evaluate $\int_0^{\infty} \frac{\cos^2(mx)}{a^2+x^2}\, dx$

```
Print["(i)"];
Clear[x]
Integrate[4 / (1 + x^2), {x, 0, 1}]

Print[];
Print["(ii)"];
Clear[ina, a, x]
ina = Integrate[x / (a^3 + x^3), x]
Simplify[D[ina, x]]

Print[];
Print["(iii)"];
Clear[ina, a, x]
ina = Integrate[(2 x^2 - x + 4) / (x^3 + 4 x), x]
Simplify[D[ina, x]]

Print[];
Print["(iv)"];
Assuming[{ a ∈ Reals, a > 0},
 Integrate[x^2 Exp[-2 a x^2], {x, -Infinity, Infinity}]]

Print[];
Print["(v)"];
Assuming[{a > 0, m ≥ 0}, Integrate[Cos[m x]^2 / (a^2 + x^2), {x, 0, Infinity}]]
```

(i)

π

(ii)

$$\frac{2\sqrt{3}\,\mathrm{ArcTan}\left[\frac{-a+2x}{\sqrt{3}\,a}\right] - 2\,\mathrm{Log}[a+x] + \mathrm{Log}\left[a^2 - a x + x^2\right]}{6a}$$

$$\frac{x}{a^3+x^3}$$

(iii)

$$-\frac{1}{2}\,\text{ArcTan}\left[\frac{x}{2}\right]+\text{Log}[x]+\frac{1}{2}\,\text{Log}\left[4+x^2\right]$$

$$\frac{4-x+2\,x^2}{4\,x+x^3}$$

(iv)

$$\frac{\sqrt{\frac{\pi}{2}}}{4\,a^{3/2}}$$

(v)

$$\frac{\left(1+e^{-2\,a\,m}\right)\,\pi}{4\,a}$$

(9) Integrate the two functions below:
f1= x^5 Sin[2 x] + x./3   in the interval [2.5,7]
f2= x^5 Exp[Sin[2 x]] + x./3   in the interval [2.5,7]

(i) Using a single command from *Mathematica*
(ii) Using a sum of small areas, as we did in class.
(iii) Using a sum of small areas, as we did in class, but avoid the commad “Sum” and instead use a do-loop.
[In (ii) and (iii), use a bin size that gives results very close to the integral done with *Mathematica*]

FUNCTION f1

```
Print["item (i)"];
Integrate[x^5 Sin[2 x] + x / 3., {x, 2.5, 7}]

Print["item (ii)"];
Clear[f1, bin, xin, xf];
f1 = x^5 Sin[2 x] + x / 3.;
bin = 0.0001;
xin = 2.5;
xf = 7.;
Sum[bin * (f1 /. x → xin + bin (k - 1) + bin / 2.), {k, 1, (xf - xin) / bin}]

Print["item (iii)"];
Clear[f1, bin, xin, xf, sumo];
f1 = x^5 Sin[2 x] + x / 3.;
bin = 0.0001;
xin = 2.5;
xf = 7.;
sumo = 0.;
Do[
 sumo = sumo + bin * (f1 /. x → xin + bin (k - 1) + bin / 2.);
 , {k, 1, (xf - xin) / bin}]
sumo
```

```
item (i)
1796.19
item (ii)
```

```
1796.19
item (iii)
1796.19
```

FUNCTION f2 -- need to use NIntegrate

```
Print["item (i)"];
NIntegrate[x^5 Exp[Sin[2 x]] + x / 3., {x, 2.5, 7}]

Print["item (ii)"];
Clear[f2, bin, xin, xf];
f2 =  x^5 Exp[Sin[2 x]] + x / 3.;
bin = 0.001;
xin = 2.5;
xf = 7.;
Sum[bin * (f2 /. x → xin + bin (k - 1) + bin / 2.), {k, 1, (xf - xin) / bin}]

Print["item (iii)"];
Clear[f2, bin, xin, xf, sumo];
f2 = x^5 Exp[Sin[2 x]] + x / 3.;
bin = 0.001;
xin = 2.5;
xf = 7.;
sumo = 0.;
Do[
 sumo = sumo + bin * (f2 /. x → xin + bin (k - 1) + bin / 2.);
 , {k, 1, (xf - xin) / bin}]
sumo
```

```
item (i)
26 648.8
item (ii)
26 648.8
item (iii)
26 648.8
```